

# Eine Absatzformatvorlage aus dem Zieldokument entfernen

Dr. Volker Thormählen, 25. Aug. 2020

## Inhalt

|     |  |   |
|-----|--|---|
| 1   | Aufgabe .....  | 2 |
| 2   | Lösung .....   | 2 |
| 1.1 | Voraussetzungen .....                                      | 2 |
| 1.2 | VBA-Prozeduren .....                                       | 2 |
| 3   | Ergebnis.....  | 3 |
| 4   | Absätze mit bestimmter Absatzformatvorlage entfernen ..... | 3 |

## Listings

|            |  |   |
|------------|--|---|
| Listing 1: | Treiber-Prozedur zum Löschen der Absatzformatvorlage 'Quellcode' .....                   | 2 |
| Listing 2: | Prozedur zum Entfernen d. Absatzformatvorlage 'Quellcode' aus d. aktuellen Dokument..... | 3 |
| Listing 3: | Alle Absätze entfernen, die mit einer best. Absatzformatvorlage formatiert wurden. ....  | 3 |

# 1 Aufgabe

In diesem Beitrag wird beschrieben, wie eine definierte Absatzformatvorlage mit VBA-Prozeduren aus einem Word-Dokument entfernt werden kann. Die Existenz eines Word-Dokuments, das die zu löschende Absatzformatvorlage enthält, wird vorausgesetzt.

Abschließend wird gezeigt, wie alle Absätze eines Word-Dokuments, die mit einer bestimmten Absatzformatvorlage (z. B. ' Quellcode') formatiert wurden, bequem mit einer VBA-Prozedur entfernt werden können.

## 2 Lösung

### 1.1 Voraussetzungen

Beim Löschen einer definierten Absatzformatvorlage (kurz: FV) muss das Folgende beachtet werden:

- Name und Typ der zu löschenden FV müssen gültig sein.
- Die zu löschende FV muss im aktiven Word-Dokument vorhanden sein.
- Die zu löschende FV darf nicht integriert sein.

### 1.2 VBA-Prozeduren

Das Löschen einer definierten Absatzformatvorlage (hier: 'Quellcode') erfolgt mit folgenden beiden VBA-Prozeduren (s. Listing 1 und Listing 2):

```
1 Sub Driver()  
2   Call FV_Entfernen(ActiveDocument, "Quellcode")  
3 End Sub
```

*Listing 1: Treiber-Prozedur zum Löschen der Absatzformatvorlage 'Quellcode'*

```

1 Sub FV_Entfernen(docTgt As Document, strFV As String)
2   ' Aufgabe: Eine vorgegebene Absatzformatvorlage (strFV) aus dem Ziel-
3   ' dokument entfernen, falls die entsprechenden Bedingungen erfüllt sind.
4   Dim objStyle As Style
5   For Each objStyle In docTgt.Styles
6     ' Prüfen, ob die Schreibweise der FV korrekt ist.
7     If objStyle.NameLocal = strFV Then
8       ' Prüfen, ob eine Absatzformatvorlage übergeben wurde.
9       If docTgt.Styles(strFV).Type = wdStyleTypeParagraph Then
10        ' Prüfen, ob die FV integriert ist.
11        If objStyle.BuiltIn = False And objStyle.Linked = False Then
12          docTgt.Styles(strFV).Delete
13          MsgBox "Die Absatzformatvorlage '" & strFV & "' wurde gelöscht!", vbExclamation
14          Exit Sub
15        Else
16          MsgBox "Achtung:" & vbCrLf & _
17            "Die Formatvorlage '" & strFV & "' kann nicht gelöscht werden!", _
18            vbCritical, "Integrierte Formatvorlage!"
19          Exit Sub
20        End If
21      Else
22        MsgBox "Achtung:" & vbCrLf & _
23          "Die Formatvorlage '" & strFV & "' ist keine Absatzformatvorlage!", _
24          vbCritical, "Kein passender Formatvorlagentyp!"
25        Exit Sub
26      End If
27    End If
28  Next objStyle
29  MsgBox "Achtung:" & vbCrLf & _
30    "Die Formatvorlage '" & strFV & "' wurde im aktuellen Word-Dokument nicht gefunden!", _
31    vbCritical, "FV existiert nicht!"
32 End Sub

```

*Listing 2: Prozedur zum Entfernen d. Absatzformatvorlage 'Quellcode' aus d. aktuellen Dokument*

### 3 Ergebnis

Das Löschen der Absatzformatvorlage 'Quellcode' bewirkt, dass alle Zeilen des Quellcodes im aktuellen Dokument mit der Basisformatvorlage 'Standard' ( $\Rightarrow$  wdStyleNormal) formatiert sind.

### 4 Absätze mit bestimmter Absatzformatvorlage entfernen

Wenn in einem Word-Dokument alle Absätze entfernt werden sollen, die mit einer bestimmten Absatzformatvorlage (z. B. 'Quellcode') formatiert wurden, kann folgende VBA-Prozedur (s. Listing 3) eingesetzt werden.

```

1 Sub RemoveSpecificParas(objDoc As Document, strStyleNm As String)
2   ' Im übergebenen Dokument (objDoc) alle Absätze mit einer
3   ' bestimmten Absatzformatvorlage (strStyleNm) entfernen.
4   Dim lngLoop As Long   ' Schleifenzähler
5   Dim lngParas As Long  ' Anzahl der Absätze im Dokument
6   lngParas = objDoc.Paragraphs.Count
7   For lngLoop = lngParas To 1 Step -1
8     With objDoc.Paragraphs(lngLoop)
9       If .Style = strStyleNm Then
10        .Range.Delete
11      End If
12    End With
13  Next lngLoop
14 End Sub

```

*Listing 3: Alle Absätze entfernen, die mit einer best. Absatzformatvorlage formatiert wurden.*